
Evil and Powerful Protocols for Elixir
Protocols are just modules!  And sometimes you just want to do
module things in protocols.  And when you're in control of both
the protocol and the module you're implementing for, colocating
the code in the module you're writing just makes sense.  This
library does this automatically for you.

  
    
  
  Usage



  
    
  
  Protocol definition


use Protoss

defprotocol MyProtocol do
  # this works like a normal protocol forward definition.
  def fun_must_be_implemented(argument)

  # delegations forward to the passed module.  Currently
  # this is not currently runtime-checked, but it will be 
  # in the future under a flag.
  #
  # call as:
  # `MyProtocol.from_json(MyStruct, ...)`
  #
  defdelegate from_json(module, argument)

  # after block separates protocol code from module code
after

  # struct protocol implementations MUST implement these
  # extra callbacks:
  @callback extra_callback() :: term

  # this can be called directly as: `MyProtocol.root_function()`
  def root_function do
    :ok
  end
end

  
    
  
  Protocol implementation


In your (struct) protocols, the implementation of MyProtocol from above
would look like this.
defmodule MyStruct do
  defstruct [:value]

  use MyProtocol

  @impl MyProtocol
  def fun_must_be_implemented(%__MODULE__{value: value}), do: value

  # note the arity change!
  @impl MyProtocol
  def from_json(%{"value" => value}), do: %__MODULE__{value: value}

  @impl MyProtocol
  def extra_callback, do: "I was required to implement this"
end

  
    
  
  Formatter changes


In order to get proper formatting, you should add the following option to your .formatter.exs
locals_without_parens: [defdelegate: 1]

  
    
  
  Installation


The package is available on hex and can be installed by adding protoss to your 
list of dependencies in mix.exs:
def deps do
  [
    {:protoss, "~> 0.2"}
  ]
end
Documentation can be found at https://hexdocs.pm/protoss.

Protoss is a utility library that allows you to define protocols implementations
which automatically delegate back to the module that is the caller of the protocol.

  
    
  
  Automatic delegation of protocol functions


The primary purpose of Protoss is to automatically delegate protocol functions.
As an example, here is standard elixir code for implementing a protocol:
defprotocol Proto do
  def unwrap(term)
end

defmodule MyStruct do
  defstruct [:value]
  defimpl Proto do
    def unwrap(term), do: term.value
  end
end
Protoss makes this easier to understand by hiding the defimpl boilerplate module.
This is a natural fit for Elixir's convention of having module functions whose
first parameter is the module's datatype.
use Protoss
defprotocol Proto do
  def unwrap(term)
end

defmodule MyStruct do
  use Proto
  defstruct [:value]
  def unwrap(term), do: term.value
end

  
    
  
  Directly delegated functions.


Protoss also allows you to define functions that get directly delegated to the
implementation module, that are then logically associated with the protocol, and
whose existence is checked by the compiler using the protocol behaviour.
A common use case would be a from_json function which reifies json data into a
struct.
Example:
use Protoss
defprotocol Proto do
  def unwrap(term)
  defdelegate from_json(module, json)
end

defmodule MyStruct do
  use Proto
  defstruct [:value]
  def unwrap(term), do: term.value

  # NOTE the arity of the function
  def from_json(%{"value" => v}), do: %__MODULE__{value: v}
end

  
    
  
  Protocol body functions


Finally, protoss allows you to write arbitrary functions in the protocol body
that aren't necessarily part of the protocol itself, using the after keyword.
This could be useful to reduce code duplication if you have a common processing 
step that must occur in conjunction with a pseudo-private protocol function, 
or even if you simply have a thematically relevant function that you would like 
to incorporate  into the same namespace.
use Protoss
defprotocol Proto do
  def _unwrap_impl(term)
after
  def unwrap(struct) do
    struct
    |> _unwrap_impl()
    |> SomeOtherModule.process()
  end
end

defmodule MyStruct do
  use Proto
  defstruct [:value]
  def _unwrap_impl(term), do: term.value
end

